

Ultracold Atoms in Optical Lattices

A thesis submitted in partial fulfillment of the
requirements for the degree of Bachelor of Science in
Physics

by

Kiatichart Chartkunchand

Dr. Andrei Derevianko, Advisor

Dr. Peter Winkler, Reader

University of Nevada, Reno

May, 2006

We recommend that the thesis
prepared under our supervision by

KIATTICHART CHARTKUNCHAND

entitled

Ultracold Atoms Trapped in Optical Lattices

be accepted in partial fulfilment of the
requirements for the degree of

Bachelor of Science

Andrei Derevianko, Ph.D., Advisor

Peter Winkler, Ph.D., Reader

Abstract

Atoms can be trapped by light in the form of optical lattices, periodic structures akin to the crystalline lattices of solid-state physics formed by the interference of laser beams. These “light crystals” share more than just superficial looks with their solid-state cousins: energy band structure and eigenfunctions in the form of Bloch functions are also present in optical lattices. In this study we numerically derive the energy bands and Bloch functions of a one-dimensional optical lattice, as well as construct an important set of functions known as the Wannier functions which give us a convenient basis for describing atoms trapped in these lattices. Several examples are given of the effect on energy band structure and Wannier functions due to changes in the parameters of the optical lattice.

Acknowledgments

I would like to thank my Advisor, Dr. Andrei Derevianko, for providing me with the necessary guidance and encouragement all throughout my research.

I would also like to thank Dr. Peter Winkler for acting as Reader for my thesis work. I especially thank him for being able to work with me on short notice.

Contents

Abstract	i
Acknowledgments	ii
List of Figures	iv
1 Introduction	1
2 Theory	2
2.1 Optical Lattices	2
2.2 Energy Bands and Bloch Functions	3
2.3 Wannier Functions	4
3 Procedures	6
3.1 Implementation of Numerov’s Method	6
3.2 Determination of Allowed Energies	8
3.3 Energies in the Harmonic Approximation	10
3.4 Construction of Bloch Functions	12
3.5 Construction of Wannier Functions	13
4 Results and Discussion	14
4.1 Energy Band Structure	14
4.2 Wannier Functions	15
4.3 Difficulties	17
5 Conclusions	18
6 Appendix A: Source Code for “lattice.f90”	19
References	25

List of Figures

1	Optical Lattice Potential	2
2	Dispersion Curve for $V_0/E_R = 10$ (first band)	14
3	Dispersion Curve for $V_0/E_R = 200$ (first band)	14
4	Dispersion Curve for Different Potential Depths	14
5	Wannier Function for $x_n = 0, V_0/E_R = 10$	15
6	Wannier Function for $x_n = 0, V_0/E_R = 200$	15
7	Wannier Function ($x_n = 0$) for Different Potential Depths	16
8	Comparison of $w(x - x_n)$ to Harmonic Oscillator Wavefunction	16

1 Introduction

Optical lattices are an ideal platform for atomic experimentation. Optical lattices have seen utilization in such diverse fields as quantum phase transitions[1] and quantum computation[2]. What makes optical lattices so useful is the nearly complete control it gives us over the system. The interaction of atoms in the lattice can easily be control by varying the parameters of the laser being used such as the laser intensity. Optical lattices are also very “clean”, that is to say that disturbances from outside the system can be kept to a minimum. In order to utilize this important tool, we must be able to describe the behavior of atoms trapped in these lattices in a thorough fashion.

This study will be mainly of a computational nature. We utilize an algorithm called Numerov’s method to numerically solve the Schrödinger equation for the potential induced by the optical lattice. From the solutions we construct the energy band structure, Bloch functions, and Wannier functions for the problem. The Bloch functions are the eigenfunctions of the problem corresponding to the allowed energies. Wannier functions are a set of functions derived from the Bloch functions which are a convenient basis to represent the atoms in the lattice due to their localized nature and property of orthogonality. To ensure the correctness of our program, we make comparisons to the Solid State Simulation (SSS) program “Bloch”. The SSS package is a suite of programs developed at Cornell University as a teaching and visualization aid in the field of solid-state physics. The program “Bloch” generates the band structure and Bloch functions for various types of potentials. We use this program as a test case for our program, making sure that we arrive at comparable results for the same input potentials before we proceed. More information on the SSS package, as well as downloads, can be found at <http://www.physics.cornell.edu/sss>.

2 Theory

2.1 Optical Lattices

Optical lattices act very much like the crystalline lattices of solid-state physics, trapping atoms at the minima of the overall potential. In this work we consider a one-dimensional optical lattice, constructed from the interference of two counter-propagating laser beams with orthogonal polarizations. Consider the two laser beams to be monochromatic plane waves of frequency ω ; then the resulting electric field is given by:

$$\mathbf{E}(x, t) = E_+(x, t)\epsilon_+e^{-i\omega t} + E_-(x, t)\epsilon_-e^{-i\omega t}.$$

Here $E_{\pm}(x, t) = E_0e^{\pm ik_L x}$, where $k_L \equiv 2\pi/\lambda$ is the wavenumber of the laser, and ϵ_{\pm} is the polarization vector. The potential $V(x)$ felt by atoms in the lattice is proportional to $|E_{\pm}(x, t)|^2$ [3]: $V(x) \propto \cos^2(kx)$. A more convenient parametrization leads us to the form of the potential we will use throughout this work: $V(x) = V_0 \sin^2(\frac{\pi}{a}x)$, where $a \equiv \lambda/2$ is the periodicity of the lattice.

Trapping of atoms within the lattice is achieved by an effect known as ‘‘Sisyphus cooling’’. We note that by having the two opposing laser beams be of orthogonal polarization to each other, the resulting polarization rotates in space: the polarization is linear at $x = 0, \lambda/4, \lambda/2, \dots$, clockwise circular at $x = \lambda/8, 5\lambda/8, \dots$, and counter-clockwise circular at $x = 3\lambda/8, 7\lambda/8, \dots$ [4]. The electric field generated by the lasers causes a shift in the energy levels of the atom due to the AC Stark effect, an electrical ana-




Figure 1: Optical Lattice Potential

logue to the Zeeman effect. Suppose an atom has two degenerate sublevels to its ground state. These levels will oscillate in space in accordance with the polarization, with one sublevel experiencing the largest shift at a point of clockwise circular polarization and the other at a point of counter-clockwise circular polarization[4]. The result is that once atoms traveling along the potential wells of one of the sublevels reach the top of those wells, they are optically pump to the bottom of the potential well of the other sublevel and lose energy in the process[4]. Eventually the atoms do not have enough energy to get over the potential hill and remain trapped around the potential minima of each lattice site.

2.2 Energy Bands and Bloch Functions

For this work, we have recourse to solving the one-dimensional, time-independent Schrödinger equation for a periodic potential:

$$\frac{-\hbar^2}{2m} \frac{d^2\psi}{dx^2} + V(x)\psi(x) = E\psi(x),$$

$$V(x) = V_0 \sin^2\left(\frac{\pi}{a}x\right),$$

where $a \equiv \frac{\lambda}{2}$ is the period of the potential. Solutions for this particular class of potentials are governed by Bloch's theorem, which states that:

$$\psi_k(x) = u_k(x)e^{ikx}, \tag{1}$$

where the function $u_k(x)$ has the periodicity of the lattice and k , the wavenumber, is real. This results in what are called Bloch functions, eigenfunctions of the Hamiltonian for this system. Thus by solving the problem in one particular cell of the potential, we can derive the wavefunction in other cells by applying Bloch's theorem.

A particular characteristic of periodic potentials is the notion of allowed and forbidden energies. Allowed energies are those that correspond to the Bloch functions. These allowed ranges of E are separated by gaps corresponding to forbidden energies, values of E which result in solutions of the same form as (1) but with imaginary values of k . Thus the forbidden energies correspond to solutions with an exponential behavior. This series of allowed and forbidden energies is called the band structure of the lattice in question.

2.3 Wannier Functions

Bloch functions give us the solution to Schrödinger's equation for the given periodic potential; it would be of great use, however, to derive a complete set of orthonormal functions which are localized about each potential minima. These set of functions are known as Wannier functions. For a given energy band l , the Wannier functions are defined by the Bloch functions as (in one dimension)[5]:

$$w_l(x - x_n) = N^{-1/2} \sum_k \exp\{-ikx_n\} \psi_{lk}(x), \quad (2)$$

where $x_n = na$ is a lattice site position with $n \in \mathbb{Z}$ and N is the number of lattice points where we consider N atoms on a string with the ends connected so as to obtain periodic boundary conditions. We show that Wannier functions are orthonormal in the following manner:

$$\begin{aligned} \int w_l^*(x - x_n) w_{l'}(x - x_{n'}) dx &= \frac{1}{N} \int \left[\sum_k e^{ikx_n} \psi_{lk}^*(x) \right] \left[\sum_{k'} e^{-ik'x_{n'}} \psi_{l'k'}(x) \right] dx \\ &= \frac{1}{N} \sum_{k,k'} e^{ikx_n} e^{-ik'x_{n'}} \int \psi_{lk}^*(x) \psi_{l'k'}(x) dx \\ &= \frac{1}{N} \sum_{k,k'} e^{ikx_n} e^{-ik'x_{n'}} \delta_{kk'} \delta_{ll'} \end{aligned}$$

$$\begin{aligned}
&= \frac{1}{N} \sum_k e^{ik(x_n - x_{n'})} \delta_{ll'} \\
&= \delta_{x_n x_{n'}} \delta_{ll'}.
\end{aligned}$$

For this study we adopt a continuous limit of the Wannier functions by noting that $\sum_k \rightarrow \frac{Na}{2\pi} \int_{-\pi/a}^{\pi/a} dk$. This comes from considering our system to be a string of N atoms. The limits of integration contain the first Brillouin zone, where the relevant values of k reside; outside of this the values repeat those within the first Brillouin zone due to the periodicity of the lattice. We also concentrate on only one particular energy band (the first band), so we drop the band index l from the Wannier function and end up with:

$$w(x - x_n) = \sqrt{\frac{a}{2\pi}} \int_{-\pi/a}^{\pi/a} \exp\{-ikx_n\} \psi_k(x) dk. \quad (3)$$

There is a certain ambiguity to the definition of the Wannier functions due to phase of the Bloch functions not being defined. To ensure that the corresponding Wannier functions are real and symmetric about each lattice point, we require that $\psi_k(x = 0)$ is real and continuous for all allowed k [6]. To ensure this amounts to multiplying the Bloch functions by a phase factor such that:

$$\psi_k(x) \rightarrow \exp\{-i \arg[\psi_k(0)]\} \psi_k(x), \quad (4)$$

where $\arg[\psi_k(0)] = \tan^{-1}\{\Im[\psi_k(0)]/\Re[\psi_k(0)]\}$.

3 Procedures

3.1 Implementation of Numerov's Method

Numerov's method[7] is an efficient numerical algorithm for solving differential equations of the same form as Schrödinger's equation, i.e. second-order linear ordinary differential equations which do not depend on the first derivative. Let $x_n \equiv x_0 + nh$ where x_0 is the starting point of integration and h is the step size. If we re-write Schrödinger's equation in the form:

$$\psi'' + \frac{2m}{\hbar^2}[E - V(x)]\psi = 0,$$

Numerov's method gives us[7]:

$$\begin{aligned} \psi_{n+1}(12 - h^2V_{n+1}) + \psi_n(-24 - 10h^2V_n) + \psi_{n-1}(12 - h^2V_{n-1}) \\ = -h^2E(\psi_{n+1} + 10\psi_n + \psi_{n-1}) + O(h^6), \end{aligned}$$

where $\psi_n \equiv \psi(x_n)$ and $V_n \equiv V(x_n)$. If we now let $k^2 \equiv E - V(x)$ and solve for ψ_{n+1} , we arrive at:

$$\psi_{n+1} = \frac{2(1 - \frac{5h^2}{12}k_n^2)\psi_n - (1 + \frac{h^2}{12}k_{n-1}^2)\psi_{n-1}}{1 + \frac{h^2}{12}k_{n+1}^2} + O(h^6), \quad (5)$$

where $k_n^2 \equiv k^2(x_n)$. This is the form of Numerov's method that is implemented in the program. From the above algorithm, we can see that each new solution is calculated from the two previous solutions. Thus to begin the process, we would require knowledge of ψ_0 and ψ_1 . This then begs the question: How do we begin the Numerov algorithm? One method would be use ψ_0 in another method such as the classic Runge-Kutta algorithm to determine ψ_1 and then continue with the Numerov

algorithm from there. The method[8] we use in this study, however, is to use an explicit expression for ψ_1 . First we re-write the Schrödinger equation in the form:

$$\frac{d^2\psi}{dx^2} = U(x)\psi,$$

where $U(x) \equiv \frac{2m}{\hbar^2}[V_0 \sin^2(\frac{\pi}{a}x) - E]$ and let $F \equiv U(x)\psi$. Then, with knowledge of ψ_0 and ψ'_0 , it can be shown that[8]:

$$\psi_1 = \frac{\psi_0(1 - \frac{U_2 h^2}{24}) + h\psi'_0(1 - \frac{U_2 h^2}{12}) + \frac{7h^2}{24}F_0 - \frac{U_2 h^4}{36}F_0}{1 - \frac{U_1 h^2}{4} + \frac{U_1 U_2 h^4}{18}}. \quad (6)$$

ψ_1 is calculated to an accuracy $O(h^5)$, which ensures that ψ_2 will be calculated $O(h^6)$ since the global error of Numerov's method is $O(h^5)$ and we only calculate ψ_1 once.

In applying Numerov's method to the problem, we first realize that since we are dealing with a second-order ordinary differential equation we can express the general solution $\psi_k(x)$ as a combination of two linearly independent solutions of the equation, $\phi_1(x)$ and $\phi_2(x)$:

$$\psi_k(x) = C_1\phi_1(x) + C_2\phi_2(x).$$

For comparison with the SSS "Bloch" program, we adopt the following boundary conditions on $\phi_1(x)$ and $\phi_2(x)$:

$$\phi_1\left(-\frac{a}{2}\right) = 1, \phi'_1\left(-\frac{a}{2}\right) = 0,$$

$$\phi_2\left(-\frac{a}{2}\right) = 0, \phi'_2\left(-\frac{a}{2}\right) = \frac{1}{a},$$

where a is the lattice constant of the potential. Numerov's method is used to determine both $\phi_1(x)$ and $\phi_2(x)$. We concentrate on a unit cell of the optical lattice centered around $x = 0$, corresponding to an integration interval of $-a/2 < x < a/2$.

3.2 Determination of Allowed Energies

Once $\phi_1(x)$ and $\phi_2(x)$ have been determined for a given energy value E , we can determine if E is an allowed energy. First let us utilize the boundary conditions defined above in $\psi_k(x)$:

$$\psi_k\left(-\frac{a}{2}\right) = C_1\phi_1\left(-\frac{a}{2}\right) + C_2\phi_2\left(-\frac{a}{2}\right) = C_1,$$

$$\psi'_k\left(-\frac{a}{2}\right) = C_1\phi'_1\left(-\frac{a}{2}\right) + C_2\phi'_2\left(-\frac{a}{2}\right) = \frac{1}{a}C_2.$$

At the endpoints of the interval, we have $\psi_k(-a/2) = u_k(-a/2)e^{-ika/2}$ and $\psi_k(a/2) = u_k(a/2)e^{ika/2}$ by application of Bloch's theorem. Since $u_k(-a/2) = u_k(a/2)$ due to the periodicity of $u_k(x)$, we end up having $\psi_k(a/2) = e^{ika}\psi_k(-a/2)$. From this we obtain the equations:

$$\psi_k\left(\frac{a}{2}\right) = e^{ika}\psi_k\left(-\frac{a}{2}\right) = e^{ika}C_1,$$

$$\psi'_k\left(\frac{a}{2}\right) = e^{ika}\psi'_k\left(-\frac{a}{2}\right) = \frac{a}{2}e^{ika}C_2.$$

To solve for the constants C_1 and C_2 , we end up with a system of linear equations in C_1 and C_2 . We can re-arrange the equations as follows:

$$\left[\phi_1\left(\frac{a}{2}\right) - e^{ika}\right]C_1 + \phi_2\left(\frac{a}{2}\right)C_2 = 0,$$

$$\phi'_1\left(\frac{a}{2}\right)C_1 + \left[\phi'_2\left(\frac{a}{2}\right) - \frac{1}{a}e^{ika}\right]C_2 = 0.$$

For a non-trivial solution to the above system, the determinant of the system must be equal to zero. Thus:

$$\det = \begin{vmatrix} \phi_1\left(\frac{a}{2}\right) - e^{ika} & \phi_2\left(\frac{a}{2}\right) \\ \phi'_1\left(\frac{a}{2}\right) & \phi'_2\left(\frac{a}{2}\right) - \frac{1}{a}e^{ika} \end{vmatrix} = 0$$

$$\begin{aligned}
&= \left[\phi_1 \left(\frac{a}{2} \right) - e^{ika} \right] \left[\phi_2' \left(\frac{a}{2} \right) - \frac{1}{a} e^{ika} \right] - \phi_1' \left(\frac{a}{2} \right) \phi_2 \left(\frac{a}{2} \right) \\
&= \phi_1 \left(\frac{a}{2} \right) \phi_2' \left(\frac{a}{2} \right) - \phi_1' \left(\frac{a}{2} \right) \phi_2 \left(\frac{a}{2} \right) + \frac{1}{a} (e^{ika})^2 - e^{ika} \phi_2' \left(\frac{a}{2} \right) - \frac{1}{a} e^{ika} \phi_1 \left(\frac{a}{2} \right).
\end{aligned}$$

We recognize the term $\phi_1 \phi_2' - \phi_1' \phi_2$ as the Wronskian for this problem, defined as:

$$W(x) = \begin{vmatrix} \phi_1(x) & \phi_2(x) \\ \phi_1'(x) & \phi_2'(x) \end{vmatrix}.$$

For a differential equation of the form $y'' + P(x)y' + Q(x)y = 0$, the Wronskian is given by Abel's formula as:

$$W(x) = W(x_0) \exp\left\{ \int_{x_0}^x P(x) dx \right\}.$$

Since the Schrödinger equation is of the required form and is independent of the first derivative, $P(x) = 0$ and thus $W(x) = W(x_0)$. This implies that the Wronskian is independent of x and we can use its known value at $W(x = -a/2)$. When we evaluate the Wronskian at $x = -a/2$, we find that $W(-a/2) = 1/a$. Plugging this into (6), we arrive at the characteristic equation:

$$\lambda^2 - \left[a\phi_2' \left(\frac{a}{2} \right) + \phi_1 \left(\frac{a}{2} \right) \right] \lambda + 1 = 0, \quad (7)$$

where $\lambda \equiv e^{ika}$. Solutions to this equation take the form of:

$$\lambda_{1,2} = e^{ika} = \frac{1}{2} \left[\phi_1 \left(\frac{a}{2} \right) + a\phi_2' \left(\frac{a}{2} \right) \right] \pm i \sqrt{1 - \frac{1}{4} \left[\phi_1 \left(\frac{a}{2} \right) + a\phi_2' \left(\frac{a}{2} \right) \right]^2}. \quad (8)$$

Defining $Q \equiv \frac{1}{2}[\phi_1(a/2) + a\phi_2'(a/2)]$, if $|Q| \leq 1$ we end up with the situation where k is real, resulting in :

$$k = \frac{1}{a} \cos^{-1} Q. \quad (9)$$

In this case the energy value E used to solve for $\phi_1(x)$ and $\phi_2(x)$ is one of the allowed energy values and will correspond to a particular Bloch function.

To determine the overall band structure of the optical lattice, we begin with an energy of 0 and work our way up to the potential height using a prescribed energy step size, testing whether the particular value yields a result of $|Q| \leq 1$.

3.3 Energies in the Harmonic Approximation

We can obtain an estimate for the energy of the first band of the lattice by approximating it with a harmonic oscillator potential. We begin by introducing a scaled form of Schrödinger's equation for the problem. Consider the Schrödinger equation for this problem:

$$\psi_k''(x) = \frac{2m}{\hbar^2} \left[V_0 \sin^2\left(\frac{\pi x}{a}\right) - E \right] \psi_k(x).$$

We introduce a dimensionless parameter $\xi \equiv \frac{x}{a}$ and re-write the equation as:

$$\psi_k''(\xi) = \frac{2ma^2}{\hbar^2} \left[V_0 \sin^2(\pi\xi) - E \right] \psi_k(\xi).$$

Next we introduce the photon recoil energy $E_R \equiv \hbar^2 k^2 / 2m$ and multiply the right-hand side of the above equation by the factor E_R/E_R :

$$\begin{aligned} \psi_k''(\xi) &= \frac{2ma^2}{\hbar^2} E_R \left[\frac{V_0}{E_R} \sin^2(\pi\xi) - \frac{E}{E_R} \right] \psi_k(\xi) \\ &= \frac{2ma^2}{\hbar^2} \left(\frac{\hbar^2 k^2}{2m} \right) \left[\frac{V_0}{E_R} \sin^2(\pi\xi) - \frac{E}{E_R} \right] \psi_k(\xi) \\ &= a^2 k^2 \left[\frac{V_0}{E_R} \sin^2(\pi\xi) - \frac{E}{E_R} \right] \psi_k(\xi) \\ &= a^2 \left(\frac{\pi}{a} \right)^2 \left[\frac{V_0}{E_R} \sin^2(\pi\xi) - \frac{E}{E_R} \right] \psi_k(\xi) \\ &= [V^S \sin^2(\pi\xi) - E^S] \psi_k(\xi). \end{aligned}$$

Here we have introduced the dimensionless parameters $V^S \equiv \pi^2 V_0 / E_R$ and $E^S \equiv \pi^2 E / E_R$. This scaling effectively gives us position in units of the lattice constant a and energy in units of the photon recoil energy E_R .

We now apply the harmonic approximation by taking a Taylor expansion of the scaled potential $V(\xi) = V^S \sin^2(\pi\xi)$ about the potential minimum at $\xi = 0$. For a lattice with a sufficiently deep potential, we can neglect terms involving powers of ξ greater than two and end up with:

$$V^S \sin^2(\pi\xi) \approx V^S \pi^2 \xi^2.$$

Equating this to the potential of a harmonic oscillator, we arrive at:

$$V^S \pi^2 \xi^2 = \frac{m\omega^2 a^2 \xi^2}{2E_R},$$

where we have taken care to scale the harmonic oscillator potential in the same manner as we did the Schrödinger equation. We now solve for ω :

$$\begin{aligned} \omega^2 &= \frac{2E_R \pi^2}{ma^2} V^S \\ &= \frac{2\pi^2}{ma^2} \left(\frac{\hbar^2 k^2}{2m} \right) V^S \\ \omega &= \frac{\pi^2 \hbar}{ma^2} \sqrt{V^S}. \end{aligned}$$

The ground-state energy of a harmonic oscillator is given by $E_0 = \frac{1}{2} \hbar \omega$. Using our expression for ω we obtain:

$$\begin{aligned} E_0 &= \frac{1}{2} \hbar \left(\frac{\pi^2 \hbar}{ma^2} \sqrt{V^S} \right) \\ &= \frac{\pi^2 \hbar^2}{2ma^2} \sqrt{V^S} \end{aligned}$$

$$= E_R \sqrt{V^S}.$$

We now have an estimate for the energy of the first band of the lattice. Letting $E_{est}^S \equiv E_0/E_R$, we finally have:

$$E_{est}^S = \sqrt{V^S} = \pi \sqrt{\frac{V_0}{E_R}}. \quad (10)$$

3.4 Construction of Bloch Functions

Now that we have the allowed values for E and k , we can determine the Bloch functions $\psi_k(x)$. Knowing the value of k helps us in determining the constants C_1 and C_2 which are involved in the construction of $\psi_k(x)$ from the two linearly independent solutions $\phi_1(x)$ and $\phi_2(x)$ of Schrödinger's equation. Setting $C_1 = 1$ for the time being (we will fix it by normalization later), we solve one of our equations from the system in the previous section for C_2 :

$$\begin{aligned} \left[\phi_1\left(\frac{a}{2}\right) - e^{ika} \right] C_1 + \phi_2\left(\frac{a}{2}\right) C_2 &= 0, \\ \left[\phi_1\left(\frac{a}{2}\right) - e^{ika} \right] + \phi_2\left(\frac{a}{2}\right) C_2 &= 0, \\ C_2 &= \frac{1}{\phi_2\left(\frac{a}{2}\right)} \left[e^{ika} - \phi_1\left(\frac{a}{2}\right) \right]. \end{aligned} \quad (11)$$

We now have our Bloch functions determined by the linear combination of $\phi_1(x)$ and $\phi_2(x)$:

$$\psi_k(x) = \phi_1(x) + C_2 \phi_2(x), \quad (12)$$

with C_2 given in (11).

There is still the matter of normalization and fixing the phase of the Bloch functions. For comparison with the SSS ‘‘Bloch’’ program, we normalize the Bloch func-

tions in the following manner:

$$N = \frac{1}{a} \int_{-a/2}^{a/2} |\psi_k(x)|^2 dx.$$

We fix the phase by requiring $\psi_k(x=0)$ to be real and continuous. Thus our final Bloch functions will take the form:

$$\psi_k(x) \rightarrow \frac{1}{\sqrt{N}} \exp\{-i \arg[\psi_k(0)]\} \psi_k(x). \quad (13)$$

3.5 Construction of Wannier Functions

With the Bloch functions determined for a particular energy band, we can now determine the Wannier functions. Recall that the Wannier functions are defined in terms of the Bloch functions by:

$$w(x - x_n) = \sqrt{\frac{a}{2\pi}} \int_{-\pi/a}^{\pi/a} \exp\{-ikx_n\} \psi_k(x) dk.$$

We begin by constructing the Wannier function for $x_n = 0$; the functions about other lattice points are given by simple translations of the one about $x_n = 0$. By the very nature of our program for constructing Bloch functions, we are limited to the unit cell about $x_n = 0$, i.e. in the interval $-a/2 < x < a/2$. Thus an initial determination of the Wannier function would give us the function in that interval. To find the Wannier function in a large enough interval where we can verify the exponential decay of the function, we determine the Wannier function about other lattice points while still looking through our restricted interval. For example, by determining $w(x - a)$, what we would see in our unit interval would be the function value of $w(x - 0)$ in the interval $-3a/2 < x < -a/2$; similarly determining the Wannier function about $x_n = -a$, we would give us the function about $x_n = 0$ in the interval $a/2 < x < 3a/2$.

4 Results and Discussion

4.1 Energy Band Structure

The band structure of the optical lattice was calculated for various potential depths. This takes the form of dispersion curves relating the energy to the wavenumber. We concentrate on the first energy band as this is where we are closest to the harmonic approximation.




Figure 2: Dispersion Curve for $V_0/E_R = 10$ (first band)




Figure 3: Dispersion Curve for $V_0/E_R = 200$ (first band)




Figure 4: Dispersion Curve for Different Potential Depths

We notice that as the potential becomes deeper, the energies tend towards a single

energy value, which should be the ground-state energy for the associated harmonic oscillator potential. Let us take the potential depth $V_0/E_R = 200$ to illustrate this observation. From the dispersion curve for this depth, we can see that the energy takes a value of roughly 41. For the corresponding harmonic energy estimate as derived previously, we have that $E_{est}^S = \pi\sqrt{V_0/E_R} \approx 44$. This would indicate that the potential depth in question is not sufficiently deep enough to approximate by a harmonic oscillator; in fact this particular potential holds roughly $200/44 \approx 4$ harmonic oscillator levels. With greater values for the potential depth, we should see closer agreement with the harmonic approximation.

4.2 Wannier Functions

From each of the band structures calculated, the corresponding Wannier function about $x_n = 0$ was also determined.




Figure 5: Wannier Function for $x_n = 0, V_0/E_R = 10$




Figure 6: Wannier Function for $x_n = 0, V_0/E_R = 200$

As the potential becomes deeper, the Wannier functions become more localized about their corresponding lattices points. The Wannier functions also seem to die off more quickly as the potential deepens. Since the energies of the first band should approach that of the harmonic oscillator ground-state energy for a sufficiently deep potential, the Wannier functions should approach the ground-state wavefunction of




Figure 7: Wannier Function ($x_n = 0$) for Different Potential Depths

the harmonic oscillator given by: $\psi_{HO}(x) = (m\omega/\pi\hbar)^{1/4} \exp\{-\frac{1}{2}\frac{m\omega}{\hbar}x^2\}$.




Figure 8: Comparison of $w(x - x_n)$ to Harmonic Oscillator Wavefunction

We see that the Wannier function in this case is noticeably different from the harmonic oscillator wavefunction. This we expect since we found that the particular potential depth this Wannier function is defined for is not deep enough to be approximated by a harmonic oscillator. That the Wannier function is less than the harmonic oscillator wavefunction makes sense in this case since it is spread over the four or so

harmonic potential levels that fit within the lattice potential.

4.3 Difficulties

Trying to reach a suitable potential depth for which the harmonic approximation could be applied proved to be difficult. As the potential depth was increased, it became much harder to isolate the first energy band. We would start around an interval centered at the harmonic approximation for the energy of the first band, but it seemed to take an exceedingly small energy step and a very long run-time to actually acquire a result. And even then only a few values for the wavenumber k would be obtained, leading to an insufficient number of Bloch functions to accurately construct the Wannier functions. Solutions to this lie more efficient coding and possibly more system resources.

Another difficulty encountered had to do with the Wannier functions themselves. Looking at the plots of the Wannier functions above we notice that as the potential deepens, the function seems to be growing in the next-nearest neighbors of the unit lattice (about $x = \pm 2a$). The expected behavior of the Wannier functions is a rapid decay of the function outside of the unit lattice. To see if this aberration in the function persists would require construction of Wannier functions at deeper potentials, but then we run into the problems stated above, namely the inability to construct accurate Wannier functions due to very small sample sizes in k . Until the first problem is result, there is not much we can do to resolve this one.

5 Conclusions

This paper has shown the necessary steps in modeling the quantum behavior of atoms trapped within optical lattices. With the band structure and Wannier functions calculated, we have the tools needed for studies of many-body effects. For example, trapping Bose-Einstein condensates within an optical lattice gives us a nice platform for experiments in quantum phase transitions. One thing we can observe is the transition between superfluid and Mott insulator states[1]. In a superfluid the atoms of the condensate are spread out over the entire lattice, with strong phase coherence throughout. In a Mott insulator, on the other hand, exact number of atoms become localized about each lattice site and there is no longer any phase coherence across the lattice. The physics of this type of system is described by the Bose-Hubbard model, which introduces a system Hamiltonian where terms representing local interaction of atoms within the same lattice site and tunneling between nearest neighbors are constructed from the Wannier functions[1].

Optical lattices are at the forefront of quantum experimentation. They allow us to easily control the periodic structure of the system and interactions between atoms in the lattices by adjusting the parameters of the lasers being used. As a consequence of using light as the building material for the lattice, we can minimize the disturbance to the fragile quantum states of the system from outside sources and keep it “clean”. Optical lattices can help us better understand the quantum world.

6 Appendix A: Source Code for "lattice.f90"

```

!-----
!Kiattichart Chartkunchand
!Spring 2006
!lattice.f90 - determines energy bands and Bloch functions for
!             given periodic potential, and then dervies the
!             associated Wannier functions
! Optical lattice potential of the form:
!       V(x) = V0*sin^2(pi*x/a)
!       where a = lambda/2
! This potential is used in the time-independent SE:
!       psi''(x) = 2m/h^2[V0*sin^2(pi*x/a) - E]psi(x)
! Introduce scaled SE:
!       z = x/a, Vs = V0/ER, Es = E/ER
!       where ER is the photon recoil energy = h^2k^2/2m
!       => psi''(z) = 2ma^2/h^2[Vs*sin^(pi*z) - Es]psi(z)
! By Bloch's theorem, solutions must satisfy the requirement:
!       psik(x) = exp{ikx}*uk(x), where uk(x+a)=uk(x)
! Search for Phi1 and Phi2 such that:
!       psik(x) = C1*Phi1(x) + C2*Phi2(x)
! units: [E] = photon recoil energy ER
!        [x] = lattice constant a
! Boundary conditions and normalization consistent with SSS "Bloch"
!subroutines:
! Numerov - performs Numerov's method for integration of SE
! Integrate - integral approx. using composite 7-point Newton-Coates rule
!-----
PROGRAM Lattice

    IMPLICIT NONE

!!!Interface subroutines:
    INTERFACE
        SUBROUTINE Numerov(E, V0, a, x0, y0, dy0, h, n, Y)
            REAL(8), INTENT(IN) :: E, V0, a, x0, y0, dy0, h
            INTEGER, INTENT(IN) :: n
            REAL(8), DIMENSION(0:), INTENT(OUT) :: Y
        END SUBROUTINE Numerov
        SUBROUTINE Integrate(F, a, b, n, soln)
            COMPLEX(8), DIMENSION(0:), INTENT(IN) :: F
            REAL(8), INTENT(IN) :: a, b
            INTEGER, INTENT(IN) :: n
            REAL(8), INTENT(OUT) :: soln
        END SUBROUTINE Integrate
    END INTERFACE

!!!Constant declarations:
    COMPLEX, PARAMETER :: i = (0,1)
    REAL, PARAMETER :: Pi = 3.14159265358979

!!!Variable declarations:

```

```

COMPLEX(8) :: C, phase
COMPLEX(8), DIMENSION(:), ALLOCATABLE :: Psi, W, Temp
COMPLEX(8), DIMENSION(:,:), ALLOCATABLE :: blochTbl
REAL(8) :: Es, dE, Vs, y10, dy10, y20, dy20, Step, Xmin, Xmax, x0, a, &
    Q, dPhi2, xi, k, soln, Emin, Emax, kmin, kmax, arg
REAL(8), DIMENSION(:), ALLOCATABLE :: Phi1, Phi2
REAL(8), DIMENSION(:,:), ALLOCATABLE :: Disp
INTEGER :: Points, Nodes, j, l, m, n, p, r, s, OpenStat, mid, numk, &
    InputStatus, knum

!!!Create files to write to:
OPEN(UNIT = 2, FILE = "dispCurve.dat", STATUS = "REPLACE", ACTION = &
    "WRITE", IOSTAT = OpenStat)
OPEN(UNIT = 4, FILE = "wannier.dat", STATUS = "REPLACE", ACTION = &
    "WRITE", IOSTAT = OpenStat)
!!!Determine allowed energies (first band only):
!!!declare lattice constant and integration range:
a = 1
Xmin = -a/2
Xmax = a/2
!!!declare grid and step size:
Points = 2100
Step = (Xmax-Xmin)/Points
mid = (Points/2)+1
!!!allocate arrays:
ALLOCATE(Phi1(0:Points), Phi2(0:Points), Psi(0:Points), Temp(0:Points))
!!!declare boundary conditions:
x0 = Xmin
y10 = 1.0
dy10 = 0.0
y20 = 0.0
dy20 = 1/a
!!!declare energy bounds, energy step, and potential strength:
Emin = 28.0
Emax = 30.0
dE = 0.001
Vs = 100.0
!!!find number of allowed energies:
WRITE(*,*), "Looking for allowed energies..."
Es = Emin
numk = 0
DO WHILE(Es <= Emax)
    !!!integrate Phi1:
    CALL Numerov(Es,Vs,a,x0,y10,dy10,Step,Points,Phi1)
    !!!integrate Phi2:
    CALL Numerov(Es,Vs,a,x0,y20,dy20,Step,Points,Phi2)
    !!!check for allowed energy:
    !!!numerical estimate of derivative Phi2 at a/2:
    dPhi2 = (11*Phi2(Points)-18*Phi2(Points-1)+9*Phi2(Points-2)- &
        2*Phi2(Points-3))/(6*Step)
    Q = 0.5*(Phi1(Points) + a*dPhi2)
    IF(ABS(Q) <= 1) THEN

```

```

        k = ACOS(Q)/a
        WRITE(2,*), k, Es
        numk = numk + 1
    END IF
    !!!increase energy by dE:
    Es = Es + dE
END DO
WRITE(*,*), numk, " allowed energies found"
CLOSE(2)
!!!write to dispCurve (-k -> k):
ALLOCATE(Disp(0:2*numk+1,0:1))
OPEN(UNIT = 2, FILE = "dispCurve.dat", STATUS = "OLD", ACTION = &
    "READ", IOSTAT = OpenStat)
n = numk - 1
m = numk
DO
    READ(2,*,IOSTAT=InputStatus), k, Es
    IF(InputStatus < 0) EXIT
    Disp(n,0) = -k
    Disp(n,1) = Es
    Disp(m,0) = k
    Disp(m,1) = Es
    n = n - 1
    m = m + 1
END DO
CLOSE(2)
OPEN(UNIT = 2, FILE = "dispCurve.dat", STATUS = "REPLACE", ACTION = &
    "WRITE", IOSTAT = OpenStat)
n = numk - 1
DO n = 0, 2*numk-1
    WRITE(2,*), Disp(n,0), Disp(n,1)
END DO
CLOSE(2)

!!!Determine Wannier function about x = 0:
WRITE(*,*), "Determining Wannier function about x = 0..."
knum = 2*numk+1
s = 19*Points + 18
!!!allocate arrays:
ALLOCATE(blochTbl(0:knum,0:Points), W(0:s))
!!!determine Bloch functions from x = -9a -> 9a:
p = 0
Xmin = -19*a/2
DO j = 9, -9, -1
    Xmax = Xmin + a
    WRITE(*,*), "*"
    !!!open file to read from:
    OPEN(UNIT = 2, FILE = "dispCurve.dat", STATUS = "OLD", ACTION = &
        "READ", IOSTAT = OpenStat)
    DO m = 0, knum
        !!!read allowed energy from file:
        READ(2,*,IOSTAT=InputStatus), k, Es
    
```

```

        IF(InputStatus < 0) EXIT
        !!!integrate Phi1:
        CALL Numerov(Es,Vs,a,x0,y10,dy10,Step,Points,Phi1)
        !!!integrate Phi2:
        CALL Numerov(Es,Vs,a,x0,y20,dy20,Step,Points,Phi2)
        !!!determine Bloch function:
        C = (EXP(i*k*a) - Phi1(Points))/Phi2(Points)
        Psi = Phi1 + C*Phi2
        !!!determine normalization and appropriate phase factor:
        Temp = ABS(Psi)**2
        CALL Integrate(Temp,Xmin,Xmax,Points,soln)
        arg = ATAN2(AIMAG(Psi(mid)),REAL(Psi(mid)))
        phase = EXP(-i*arg)
        Psi = a*phase*Psi/SQRT(soln)
        DO n = 0, Points
            blochTbl(m,n) = EXP(-i*k*j*a)*Psi(n)
        END DO
    END DO
    DEALLOCATE(Temp)
    ALLOCATE(Temp(0:knum))
    !!!calculate Wannier function:
    r = 0
    kmin = -Pi/a
    kmax = Pi/a
    DO n = p*Points+p, (p+1)*Points+p
        DO m = 0, knum
            Temp(m) = blochTbl(m,r)
        END DO
        CALL Integrate(Temp,kmin,kmax,knum,soln)
        W(n) = soln
        r = r + 1
    END DO
    DEALLOCATE(Temp)
    ALLOCATE(Temp(0:Points))
    p = p + 1
    CLOSE(2)
    Xmin = Xmax
END DO
WRITE(*,*), "Wannier function found!"
Xmin = -19*a/2
Xmax = 19*a/2
Step = (Xmax-Xmin)/s
x0 = Xmin
DO n = 0, s
    xi = x0 + n*Step
    WRITE(4,*), xi, REAL(W(n))
END DO

```

END PROGRAM Lattice

```

!-----
!subroutine Numerov: implements Numerov's method
!for y'' = U(x) + V(x)y:
!  let F = U(x) + V(x)y
!  for SE -
!      in SSS "Bloch": y'' = (2m/h^2)[V(x) - E]y
!      in scaled units: y'' = (2ma^2/h^2)[V(x) - E]y
! => U(x) = 0, V(x) = (2ma^2/h^2)[V(x) - E]
! => F = (2ma^2/h^2)[V(x) - E]y
!  for comparison with SSS "Bloch": 2m/h^2 = 0.262465
!input:
!  E - estimate of eigenenergy
!  V0 - strength of potential
!  a - lattice constant
!  x0 - starting point of integration
!  y0 - value of wavefunction at starting point of integration
!  dy0 - derivative of wavefunction at starting point of integration
!  h - step size
!  n - number of grid points
!  Y - array holding values of wavefunction at grid points
!output:
!  Y - array representing solution to SE for given E
!***equation for y1 derived in:
!  "Getting Started With Numerov's Method"
!  J.L.M. Quiroz Gonzalez and D. Thompson
!  Computers in Physics, Vol. 11, No. 5, Sep/Oct 1997
!-----
SUBROUTINE Numerov(E, V0, a, x0, y0, dy0, h, n, Y)

    IMPLICIT NONE

    REAL, PARAMETER :: Pi = 3.14159265358979
    REAL(8), INTENT(IN) :: E, V0, a, x0, y0, dy0, h
    INTEGER, INTENT(IN) :: n
    REAL(8), DIMENSION(0:), INTENT(OUT) :: Y
    REAL(8) :: F, k2(0:n), C1, C2, C3, C4, C5, xi, V1, V2, V3
    INTEGER :: i

    C1 = h**2 / 24
    C2 = h**2 / 12
    C3 = h**4 / 36
    C4 = h**2 / 4
    C5 = h**4 / 18
    V1 = V0*SIN(Pi*x0/a)**2 - E
    V2 = V0*SIN(Pi*(x0+h)/a)**2 - E
    V3 = V0*SIN(Pi*(x0+2*h)/a)**2 - E
    Y(0) = y0
    F = V1*Y(0)
    !!!compute value for y1:
    Y(1) = (Y(0)*(1-C1*V3) + h*dy0*(1-C2*V3) + 7*C1*F - C3*F*V3)/ &
        (1 - C4*V2 + C5*V2*V3)
    !!!implement Numerov's method:

```

```

DO i = 0, n
    xi = x0 + i*h
    k2(i) = E-V0*SIN(Pi*xi/a)**2
END DO
DO i = 1, n-1
    Y(i+1) = (2*(1-5*C2*k2(i))*Y(i) - (1+C2*k2(i-1))*Y(i-1))/ &
        (1+C2*k2(i+1))
END DO

END SUBROUTINE Numerov

!-----
!subroutine Integrate: integral approximation using composite
!                      7-point Newton-Coates rule
!input:
!  soln - variable to hold value of integration
!  a - lower limit of integration
!  b - upper limit of integration
!  n - number of points in integration interval
!  F - array containing function values to integrate
!output:
!  soln - value of integration
!-----
SUBROUTINE Integrate(F, a, b, n, soln)

    IMPLICIT NONE

    COMPLEX(8), DIMENSION(0:), INTENT(IN) :: F
    REAL(8), INTENT(IN) :: a, b
    INTEGER, INTENT(IN) :: n
    REAL(8), INTENT(OUT) :: soln
    REAL(8) :: sum, h
    INTEGER :: i

    h = (b-a)/n
    sum = 0
    DO i = 1, n/7
        sum = sum + 41*F(7*i-7) + 216*F(7*i-6) + 27*F(7*i-5) + 272*F(7*i-4) + &
            27*F(7*i-3) + 216*F(7*i-2) + 41*F(7*i-1)
    END DO
    soln = (h/140)*sum

END SUBROUTINE Integrate

```

References

- [1] D. Jaksch *et al.*, Phys. Rev. Lett. **81**, 3108 (1998)
- [2] A. Derevianko and C. C. Cannon, Phys. Rev. A **70**, 62319 (2004)
- [3] D. Jaksch and P. Zoller, *The Cold Atom Hubbard Toolbox* (2004) URL <http://www.citebase.org/cgi-bin/citations?id=oai:arXiv.org:cond-mat/0410614>
- [4] J. Dalibard and C. Cohen-Tannoudji, J. Opt. Soc. Am. B **6**, 2023 (1989)
- [5] G. Wannier, Phys. Rev. **52**, 191 (1937)
- [6] W. Kohn, Phys. Rev. **115**, 809 (1959)
- [7] V. Fack and G. Vanden Berghe, J. Phys. A: Math. Gen. **20**, 4153 (1987)
- [8] J. L. M. Quiroz González and D. Thompson, Computers in Physics **11**, 514 (1997)